
GALAHAD SORT
USER DOCUMENTATION GALAHAD Optimization Library version 2.5

1 SUMMARY

GALAHAD SORT is a suite of Fortran procedures for sorting and permuting. It includes two algorithms (heapsort and
quicksort) to sort integer and/or real vectors, another to reorder a sparse matrix from co-ordinate to row format, and a
further three tools for in-place permutation and permutation inversion.

ATTRIBUTES — Versions: GALAHAD SORT single, GALAHAD SORT double. Date: March 2002.Origin: N. I.
M. Gould, Rutherford Appleton Laboratory, and Ph. L. Toint,University of Namur, Belgium.Language: Fortran 95
+ TR 15581 or Fortran 2003.

2 HOW TO USE THE PACKAGE

Access to the package requires aUSE statement such as

Single precision version

USE GALAHAD SORT single

Double precision version

USE GALAHAD SORT double

2.1 Argument lists and calling sequences

There are seven procedures that may be called by the user.

1. The subroutineSORT inplace invert is used to invert a permutation vector without resorting to extra storage.

2. The subroutineSORT inplace permute is used to apply a given permutation to an integer vector and,optionally,
to a real vector, without resorting to extra storage.

3. The subroutineSORT inverse permute is used to apply the inverse of a given permutation to an integer vector
and, optionally, to a real vector, without resorting to extra storage.

4. The subroutineSORT reorder by rows is used to reorder a sparse matrix from arbitary co-ordinateorder to
row order, that is so that the entries for rowi appear directly before those for rowi+1.

5. The subroutineSORT quicksort is used to sort a given integer/real vectorin ascending order, optionally
applying the same permutation to integer and/or to a real vector(s). It uses the “quicksort” algorithm (see
Section 4).

6. The subroutineSORT heapsort build is used to initialize a procedure to sort a vector of real numbers using
the “heapsort” algorithm (see Section 4).

7. The subroutineSORT heapsort smallest is used to find, possibly repeatedly, the smallest componentof a real
vector to whichSORT heapsort build has been previously applied (see Section 4). Successive application of
this subroutine therefore results in sorting the initial vector in ascending order.

Note that the subroutinesSORT heapsort build andSORT heapsort smallest are particularly appropriate if it is
not known in advance how many successive smallest components of the vector will be required as the heapsort method
is able to calculate thek+1-st smallest component efficiently once it has determined the firstk smallest components. If
a complete sort is required, the Quicksort algorithm,SORT quicksort may be preferred. Both methods are guaranteed
to sort alln numbers inO(n logn) operations.

We use square brackets[ ] to indicateOPTIONAL arguments.

2.1.1 In-place inversion of a permutation

A permutationp of sizen is a vector ofn integers ranging from 1 ton, each integer in this range occurring exactly
once. Its inverse is another permutationq, also of sizen, such thatq(p(i)) = i for all i = 1, . . . ,n. Inverting a given
permutation without resorting to extra storage is done as follows:

CALL SORT inplace invert ( n, p )

n is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set by the user ton, the size of the permu-
tation to be inverted.Restriction: n> 0.

p is a rank-oneINTENT(INOUT) array argument of dimension at leastn and type either defaultINTEGER or default
REAL (double precision inGALAHAD SORT double), that must be set by the user on input so that itsi-th com-
ponent contains the integerp(i). On exit, the elements ofp will have overwritten by those ofq, the inverse of
p.

2.1.2 Applying a given permutation in place

Applying a given permutationp to a vectorx consists in modifying the vectorx such that itsi-th component appears
(after applying the permutation) in componentp(i). This is done without resorting to extra storage as follows:

CALL SORT inplace permute ( n, p [, x] [, ix] [, iy] )

n is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set by the user ton, the size of the permu-
tation to be applied.Restriction: n> 0.

p is a rank-oneINTENT(INOUT) array argument of dimension at leastn and type either defaultINTEGER or default
REAL (double precision inGALAHAD SORT double), that must be set by the user on input so that itsi-th compo-
nent contains the integerp(i), that is thei-th component of the permutation one wishes to apply. It is unchanged
on exit.

x is an optional rank-oneINTENT(INOUT) array argument of dimension at leastn and typeREAL (double precision in
GALAHAD SORT double), whosen first components must be set by the user. Ifx is present, the componentx(i)
will have been replaced byx(p(i)) on exit.

ix is an optional rank-oneINTENT(INOUT) array argument of dimension at leastn and typeINTEGER, whosen fisrt
components must be set by the user. Ifix is present, the componentix(i) will have been replaced byix(p(i))
on exit.

iy is an optional rank-oneINTENT(INOUT) array argument of dimension at leastn and typeINTEGER, whosen first
components must be set by the user. Ifiy is present, the componentiy(i) will have been replaced byiy(p(i))
on exit.

2.1.3 Applying the inverse of a given permutation in place

Applying a the inverse of a given permutationp to a vectorx consists in modifying the vectorx such that itsi-
th component appears (after applying the procedure) in componentq(i), whereq is the inverse ofp. Equivalently,
this can be seen as modifying the vectorx such that itsp(i)-th component appears (after applying the procedure) in
componenti. This is done without resorting to extra storage as follows:

CALL SORT inverse permute ( n, p [, x] [, ix] )

n is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set by the user ton, the size of the permu-
tation whose inverse is to be applied.Restriction: n > 0.

p is a rank-oneINTENT(INOUT) array argument of dimension at leastn and type either defaultINTEGER or default
REAL (double precision inGALAHAD SORT double), that must be set by the user on input so that itsi-th compo-
nent contains the integerp(i), that is thei-th component of the permutation whose inverse is to be applied. It is
unchanged on exit.

x is an optional rank-oneINTENT(INOUT) array argument of dimension at leastn and typeREAL (double precision in
GALAHAD SORT double), whosen first components must be set by the user. Ifx is present, the componentx(i)
will have been replaced byx(p(i)) on exit.

ix is an optional rank-oneINTENT(INOUT) array argument of dimension at leastn and typeINTEGER, whosen first
components must be set by the user. Ifix is present, the componentix(i) will have been replaced byix(p(i))
on exit.

2.1.4 Reordering a sparse matrix from co-ordinate to row order

The matrixA is reordered from co-ordinate to row order as follows:

CALL SORT_reorder_by_rows( nr, nc, nnz, A_row, A_col, la, A_val, A_ptr, lptr, &
IW, liw, error, warning, inform )

nr is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set by the user to the number of rows in
A. Restriction: nr> 0.

nc is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set by the user to the number of columns
in A. Restriction: nc> 0.

nnz is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set by the user to the number of nonzeros
in A. Restriction: nnz> 0.

A row is a rank-oneINTENT(INOUT) array argument of type defaultINTEGER and lengthla. On entry,A row(k),
k = 1, . . . , nnz give the row indices ofA. On exit,A row will have been reordered, butA row(k) will still be the
row index corresponding to the entry with column indexA col(k).

A col is a rank-oneINTENT(INOUT) array argument of type defaultINTEGER and lengthla. On entry,A col(k),
k = 1, . . . , nnz give the column indices ofA. On exit,A col will have been reordered so that entries in rowi
appear directly before those in rowi+1 for i = 1, . . . , nr−1.

la is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set by the user to the actual dimension
of the arraysA row, A col andA val Restriction: la ≥ nnz.

A val is a rank-oneINTENT(INOUT) array argument of type defaultREAL (double precision inGALAHAD SORT -
double) of lengthla. On entry,A val(k), k = 1, . . . , nnz give the values ofA. On exit,A val will have been
reordered so that entries in rowi appear directly before those in rowi+1 for i = 1, . . . , nr−1 and correspond to
those inA row andA col.

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .
For any commercial application, a separate license must be signed.

GALAHAD SORT (May 8, 2013) 3



SORT GALAHAD

A ptr is a rank-oneINTENT(OUT) array argument of type defaultINTEGER and lengthlptr. On exit,A ptr(i), i =
1, . . . , nr give the starting addresses for the entries inA row/A col/A val in row i, whileA ptr(nr+1) gives the
index of the first non-occupied component ofA.

lptr is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set by the user to the actual dimension
of A ptr. Restriction: lptr ≥ nr + 1.

IW is a rank-oneINTENT(OUT) array argument of type defaultINTEGER and lengthliw that is used for workspace.

liw is a scalarINTENT(IN) argument of type defaultINTEGER, that gives the actual dimension ofIW. Restriction:
liw ≥ MAX(nr,nc) + 1.

error is a scalarINTENT(IN) argument of type defaultINTEGER, that holds the stream number for error messages.
Error messages will only occur iferror > 0.

warning is a scalarINTENT(IN) argument of type defaultINTEGER, that holds the stream number for warning mes-
sages. Warning messages will only occur ifwarning > 0.

inform is a scalarINTENT(OUT) argument of type defaultINTEGER. A successful call toSORT reorder by rows is
indicated wheninform has the value 0 on exit. For other return values ofinform, see Section 2.2.

2.1.5 Quicksort

The vectorx is sorted in acending order as follows:

CALL SORT quicksort ( n, x, inform [, ix ] [, rx] )

n is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set by the user ton, the number of entries
of x that are to be sorted.Restriction: n> 0 andn < 232.

x is a rank-oneINTENT(INOUT) array argument of dimension at leastn and type either defaultINTEGER or default
REAL (double precision inGALAHAD SORT double), whose firstn components must be set by the user on input.
On successful return, these components will have been sorted to ascending order.

inform is a scalarINTENT(OUT) argument of type defaultINTEGER. A successful call toSORT quicksort is indicated
wheninform has the value 0 on exit. For other return values ofinform, see Section 2.2.

ix is an optional rank-oneINTENT(INOUT) array argument of dimension at leastn and type defaultINTEGER. If ix
is present, exactly the same permutation is applied to the components ofix as to the components ofx. For
example, the inverse permutation will be provided ifix(i) is set toi, for i = 1, . . . ,n on entry.

rx is an optional rank-oneINTENT(INOUT) array argument of dimension at leastn and type defaultREAL (double
precision inGALAHAD SORT double). If rx is present, exactly the same permutation is applied to the components
of rx as to the components ofx.

2.1.6 Heapsort

Building the initial heap The initial heap is constructed as follows:

CALL SORT heapsort build ( n, x, inform [, ix ] )

n is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set by the user ton, the number of entries
of A that are to be (partially) sorted.Restriction: n> 0.

x is a rank-oneINTENT(INOUT) array argument of dimension at leastn and type either defaultINTEGER or default
REAL (double precision inGALAHAD SORT double), whose firstn components must be set by the user on input.
On successful return, the elements ofx will have been permuted so that they form a heap.

inform is a scalarINTENT(OUT) argument of type defaultINTEGER. A successful call toSORT heapsort build is
indicated wheninform has the value 0 on exit. For other return values ofinform, see Section 2.2.

ix is an optional rank-oneINTENT(INOUT) array argument of dimension at leastn and type defaultINTEGER. If ix
is present, exactly the same permutation is applied to the components ofix as to the components ofx. For
example, the inverse permutation will be provided ifix(i) is set toi, for i = 1, . . . ,n on entry.

Finding the smallest entry in the current heap To find the smallest entry in a given heap, to place this entry at the
end of the list of entries in the heap and to form a new heap withthe remaining entries:

CALL SORT heapsort smallest ( m, x, inform [, ix ] )

m is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set by the user tom, the number of entries
of A that lie on the heap on entry.Restriction: m> 0.

x is a rank-oneINTENT(INOUT) array argument of dimension at leastm and type either defaultINTEGER or default
REAL (double precision inGALAHAD SORT double) whose firstm components must be set by the user on input
so that they form a heap. In practice, this normally means that they have been placed on a heap by a previous
call toSORT heapsort build or SORT heapsort smallest. On output, the smallest of the firstm components
of x will have been moved to positionx(m) and the remaining components will now occupy locations 1,2, ....
m-1 of x and will again form a heap.

inform is a scalarINTENT(OUT) argument of type defaultINTEGER. A successful call toSORT heapsort smallest
is indicated wheninform has the value 0 on exit. For other return values ofinform, see Section 2.2.

ix is an optional rank-oneINTENT(INOUT) array argument of dimension at leastm and type defaultINTEGER. If ix is
present, exactly the same permutation is applied to the components ofix as to the components ofx.

Finding the k smallest components of a set ofn elements To find thek smallest components of a set,{x1,x2, ...,xn},
of n elements, the user should firstly callSORT heapsort build with n = n andx1 to xn stored inx(1) to x(n). This
places the components ofx on a heap. This should then be followed byk calls ofSORT heapsort smallest, with m
= n− i+1 for i = 1, ...,k. The requiredk smallest values, in increasing order, will now occupy positionsn− i+1 of
x for i = 1, ...,k.

2.2 Warning and error messages

A positive value ofinform on exit fromSORT reorder by rows, SORT quicksort, SORT heapsort build or
SORT heapsort smallest indicates that an input error has occurred. The other arguments will not have been altered.
The only possible values are:

1. One or more of the restrictionsnr > 0, nc > 0, nnz > 0 (SORT reorder by rows), n > 0 (SORT quicksort,
SORT heapsort build) or m > 0 (SORT heapsort smallest) has been violated.

2. One of the restrictionsla ≥ nnz (SORT reorder by rows) or n < 232 (SORT quicksort) has been violated.

3. The restrictionliw ≥ MAX(nr,nc)+1 (SORT reorder by rows) has been violated.

4. The restrictionlptr ≥ nr+1 (SORT reorder by rows) has been violated.

5. All of the entries input inA row andA col are out of range.

A negative value ofinform on exit fromSORT reorder by rows indicates thatA has been successfully reordered,
but that a a warning condition has occurred. The only possible values are:

-1. There were duplicate input entries, which have been summed.

-2. There were input row entries out of range, which have beenignored.

-3. There were input column entries out of range, which have been ignored.

3 GENERAL INFORMATION

Use of common: None.

Workspace: None.

Other routines called directly: None.

Other modules used directly: None.

Input/output: None.

Restrictions: nr > 0, nc > 0, nnz > 0, la ≥ nnz, liw ≥ MAX(nr,nc)+1, lptr ≥ nr+1, (SORT reorder by rows),
n> 0 (SORT quicksort andSORT heapsort build), m> 0 (SORT heapsort smallest) andn< 232

(SORT quicksort)

Portability: ISO Fortran 95 + TR 15581 or Fortran 2003. The package is thread-safe.

4 METHOD

4.1 Quicksort

The quicksort method is due to C. A. R. Hoare (Computer Journal, 5 (1962), 10-15).

The idea is to take one component of the vector to sort, sayx1, and to move it to the final position it should occupy
in the sorted vector, say positionp. While determining this final position, the other components are also rearranged
so that there will be none with smaller value to the left of position p and none with larger value to the right. Thus the
original sorting problem is transformed into the two disjoint subproblems of sorting the firstp−1 and the lastn− p
components of the resulting vector. The same technique is then applied recursively to each of these subproblems. The
method is likely to sort the vectorx in O(n logn) operations, but may require as many asO(n2) operations in extreme
cases.

4.2 Heapsort

The heapsort method is due to J. W. J. Williams (Algorithm 232, Communications of the ACM7 (1964), 347-348).
SubroutineSORT heapsort build is a partial amalgamation of Williams’ Algol proceduressetheap andinheap while
SORT heapsort smallest is based upon his proceduresoutheap andswopheap.

The elements of the set{x1,x2, ...,xn} are first allocated to the nodes of a heap. A heap is a binary tree in which the
element at each parent node has a numerical value as small as or smaller than the elements at its two children. The
smallest value is thus placed at the root of the tree. This value is now removed from the heap and a subset of the
remaining elements interchanged until a new, smaller, heapis constructed. The smallest value of the new heap is now
at the root and may be removed as described above. The elements of the initial set may thus be arranged in order
of increasing size, thei-th largest element of the array being found in thei-th sweep of the method. The method is
guaranteed to sort alln numbers inO(n logn) operations.

5 EXAMPLE OF USE

The following example is a somewhat unnatural sequence of operations, but illustrates the use of theSORT tools. It
uses the data vector

x = {x1,x2, ...,x20}= {−5,−7, 2, 9, 0,−3, 3, 5,−2,−6, 8, 7,−1,−8, 10,−4, 6,−9, 1, 4}.

Suppose now that we wish to perform the following successfuloperations:

1. sort the components ofx in ascending order and compute the associated inverse permutation,

2. apply this permutation to the resulting vector (in order to recover its original ordering),

3. restore the permutation to the identity by sorting its components in ascending order,

4. find the 12 smallest components ofx and the associated inverse permutation,

5. inverse this permutation (which yields the permutation used to sort the 12 smallest components),

6. apply to the permutedx the inverse of the this latest permutation (thus recoveringits original ordering again).

Then we may use the following code

PROGRAM GALAHAD_SORT_EXAMPLE
USE GALAHAD_SORT_double ! double precision version
IMPLICIT NONE
INTEGER, PARAMETER :: wp = KIND( 1.0D+0 ) ! set precision
INTEGER, PARAMETER :: n = 20
INTEGER :: i, m, inform
INTEGER :: p( n )
REAL ( KIND = wp ) :: x( n )
x = (/ -5.0, -7.0, 2.0, 9.0, 0.0, -3.0, 3.0, 5.0, -2.0, -6.0, &

8.0, 7.0, -1.0, -8.0, 10.0, -4.0, 6.0, -9.0, 1.0, 4.0 /) ! values
p = (/ 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, &

15, 16, 17, 18, 19, 20 /) ! indices
! write the initial data

WRITE( 6, "( /’ The vector x is’ / 2( 10 ( F5.1, 2X ) / ) ) " ) x( 1:n )
WRITE( 6, "( ’ The permutation is’ / 20 ( 2X, I2 ) / ) " ) p( 1:n )

! sort x and obtain the inverse permutation
WRITE( 6, "( ’ Sort x in ascending order’ / )" )
CALL SORT_quicksort( n, x, inform, p )
WRITE( 6, "( ’ The vector x is now’ / 2( 10 ( F5.1, 2X ) / ) ) " ) x( 1:n )
WRITE( 6, "( ’ The permutation is now’ / 20 ( 2X, I2 ) / ) " ) p( 1:n )

! apply the inverse permutation to x
WRITE( 6, "( ’ Apply the permutation to x’ / )" )
CALL SORT_inplace_permute( n, p, x )
WRITE( 6, "( ’ The vector x is now’ / 2( 10 ( F5.1, 2X ) / ) ) " ) x( 1:n )

! restore the identity permutation
WRITE( 6, "( ’ Restore the identity permutation by sorting’ / )" )
CALL SORT_quicksort( n, p, inform )
WRITE( 6, "( ’ The permutation is now’ / 20 ( 2X, I2 ) / ) " ) p( 1:n )

! get the 12 smallest components and the associated inverse permutation
WRITE( 6, "( ’ Get the 12 smallest components of x’ / )" )

CALL SORT_heapsort_build( n, x, inform, p ) ! Build the heap
DO i = 1, 12
m = n - i + 1
CALL SORT_heapsort_smallest( m, x, inform, p ) ! Reorder the variables
WRITE( 6, "( ’ The ’, I2, ’-th(-st) smallest value, x(’, I2, ’) is ’, &
& F5.1 ) " ) i, p( m ), x( m )
END DO
WRITE( 6, "( / ’ The permutation is now’ / 20 ( 2X, I2 ) / ) " ) p( 1:n )

! compute the direct permutation in p
WRITE( 6, "( ’ Compute the inverse of this permutation’ / )" )
CALL SORT_inplace_invert( n, p )
WRITE( 6, "( ’ The permutation is now’ / 20 ( 2X, I2 ) / ) " ) p( 1:n )

! apply inverse permutation
WRITE( 6, "( ’ Apply the resulting permutation to x’ / )" )
CALL SORT_inverse_permute( n, p, x )
WRITE( 6, "( ’ The final vector is’ / 2( 10 ( F5.1, 2X ) / ) ) " ) x( 1:n )
STOP
END PROGRAM GALAHAD_SORT_EXAMPLE

This produces the following output:

The vector x is
-5.0 -7.0 2.0 9.0 0.0 -3.0 3.0 5.0 -2.0 -6.0
8.0 7.0 -1.0 -8.0 10.0 -4.0 6.0 -9.0 1.0 4.0

The permutation is
1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20

Sort x in ascending order

The vector x is now
-9.0 -8.0 -7.0 -6.0 -5.0 -4.0 -3.0 -2.0 -1.0 0.0
1.0 2.0 3.0 4.0 5.0 6.0 7.0 8.0 9.0 10.0

The permutation is now
18 14 2 10 1 16 6 9 13 5 19 3 7 20 8 17 12 11 4 15

Apply the permutation to x

The vector x is now
-5.0 -7.0 2.0 9.0 0.0 -3.0 3.0 5.0 -2.0 -6.0
8.0 7.0 -1.0 -8.0 10.0 -4.0 6.0 -9.0 1.0 4.0

Restore the identity permutation by sorting

The permutation is now
1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20

Get the 12 smallest components of x

The 1-th(-st) smallest value, x(18) is -9.0
The 2-th(-st) smallest value, x(14) is -8.0
The 3-th(-st) smallest value, x( 2) is -7.0
The 4-th(-st) smallest value, x(10) is -6.0
The 5-th(-st) smallest value, x( 1) is -5.0
The 6-th(-st) smallest value, x(16) is -4.0
The 7-th(-st) smallest value, x( 6) is -3.0
The 8-th(-st) smallest value, x( 9) is -2.0
The 9-th(-st) smallest value, x(13) is -1.0
The 10-th(-st) smallest value, x( 5) is 0.0
The 11-th(-st) smallest value, x(19) is 1.0
The 12-th(-st) smallest value, x( 3) is 2.0

The permutation is now
7 20 12 17 8 4 11 15 3 19 5 13 9 6 16 1 10 2 14 18

Compute the inverse of this permutation

The permutation is now
16 18 9 6 11 14 1 5 13 17 7 3 12 19 8 15 4 20 10 2

Apply the resulting permutation to x

The final vector is
-5.0 -7.0 2.0 9.0 0.0 -3.0 3.0 5.0 -2.0 -6.0
8.0 7.0 -1.0 -8.0 10.0 -4.0 6.0 -9.0 1.0 4.0

