
GALAHAD GLRT
USER DOCUMENTATION GALAHAD Optimization Library version 2.5

1 SUMMARY

Given realn by n symmetric matricesH andM (with M positive definite), realn vectorsc ando, and scalarsσ ≥ 0,
ε ≥ 0 and f0, this package finds anapproximate minimizer of the regularised objective function

1
p σ[‖x+o‖2

M + ε]p/2+ 1
2xT Hx + cT x+ f0,

where‖v‖M =
√

vT Mv is theM -norm ofv. This problem commonly occurs as a subproblem in nonlinear optimization
calculations involving cubic regularisation. The method may be suitable for largen as no factorization ofH is required.
Reverse communication is used to obtain matrix-vector products of the formHz andM−1z.

ATTRIBUTES — Versions: GALAHAD GLRT single, GALAHAD GLRT double. Uses: GALAHAD SYMBOLS, GALA-
HAD SPACE, GALAHAD RAND, GALAHAD NORMS, GALAHAD GLTR, GALAHAD ROOTS, GALAHAD SPECFILE, *TTRF. Date:
November 2007.Origin: N. I. M. Gould, Oxford University and Rutherford Appleton Laboratory. Language:
Fortran 95 + TR 15581 or Fortran 2003.

2 HOW TO USE THE PACKAGE

Access to the package requires aUSE statement such as

Single precision version

USE GALAHAD GLRT single

Double precision version

USE GALAHAD GLRT double

If it is required to use both modules at the same time, the derived typesGLRT control type, GLRT inform type,
GLRT data type, (Section 2.1) and the subroutinesGLRT initialize, GLRT solve, GLRT terminate (Section 2.2)
andGLRT read specfile (Section 2.6) must be renamed on one of theUSE statements.

2.1 The derived data types

Three derived data types are accessible from the package.

2.1.1 The derived data type for holding control parameters

The derived data typeGLRT control type is used to hold controlling data. Default values may be obtained by calling
GLRT initialize (see Section 2.2.1). The components ofGLRT control type are:

error is a scalar variable of type defaultINTEGER, that holds the stream number for error messages. Printing of error
messages inGLRT solve andGLRT terminate is suppressed iferror≤ 0. The default iserror = 6.

out is a scalar variable of type defaultINTEGER, that holds the stream number for informational messages. Printing
of informational messages inGLRT solve is suppressed ifout< 0. The default isout = 6.

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .

print level is a scalar variable of type defaultINTEGER, that is used to control the amount of informational output
which is required. No informational output will occur ifprint level≤ 0. If print level = 1 a single line
of output will be produced for each iteration of the process.If print level ≥ 2 this output will be increased
to provide significant detail of each iteration. The defaultis print level = 0.

itmax is a scalar variable of type defaultINTEGER, that holds the maximum number of iterations which will be
allowed inGLRT solve. If itmax is set to a negative number, it will be reset byGLRT solve to n. The default is
itmax = -1.

extra vectors is a scalar variable of type defaultINTEGER, that specifies the number of additional vectors of length
n that will be allocated to try to speed up the computation during the second pass. The default isextra vectors
= 0.

stopping rule is a scalar variable of type defaultINTEGER, that flags the stopping rule to be used (seestop relative
andstop absolute below). Appropriate values are in the range[0,2], and any value outside this range will be
interpreted as 0. The default isstopping rule = 0.

freq is a scalar variable of type defaultINTEGER, that defines the frequency at which the tridiagonal subproblem will
be solved. Specifically, the subproblem will be solved on iterationsk = 1+ freq ∗i, for i = 1,2, . . . .. If itmax
is≤ 1, it will be reset byGLRT solve to 1. The default isfreq = 1.

unitm is a scalar variable of type defaultLOGICAL, that must be set.TRUE. if the matrixM is the identity matrix, and
.FALSE. otherwise. The default isunitm = .TRUE..

space critical is a scalar variable of type defaultLOGICAL, that may be set.TRUE. if the user wishes the package
to allocate as little internal storage as possible, and.FALSE. otherwise. The package may be more efficient if
space critical is set.FALSE.. The default isspace critical = .FALSE..

deallocate error fatal is a scalar variable of type defaultLOGICAL, that may be set.TRUE. if the user wishes the
package to return to the user in the unlikely event that an internal array deallocation fails, and.FALSE. if the
package should be allowed to try to continue. The default isdeallocate error fatal = .FALSE..

stop relative andstop absolute are scalar variables of typeREAL (double precision inGALAHAD GLRT double),
that holds the relative and absolute convergence tolerances (see Section 4). The computed solutionx is ac-
cepted byGLRT solve if the computed value of‖Hx + λMx + c‖M−1 is less than or equal to max(ν‖c‖M−1∗
stop relative, stop absolute), whereλ = σ[‖x+ o‖2

M + ε]p/2−1 andν depends on the stopping rule se-
lected bystopping rule: for stopping rule = 0, ν = 1, for stopping rule = 1, ν = min(1,‖x‖), and for
stopping rule = 2,ν = min(1,‖x‖/max(1,σ)). The defaults arestop relative =

√
u andstop absolute

= 0.0, whereu is EPSILON(1.0) (EPSILON(1.0D0) in GALAHAD GLRT double).

fraction opt is a scalar variable of type defaultREAL (double precision inGALAHAD GLRT double), that specifies
the fraction of the optimal value which is to be considered acceptable by the algorithm. A negative value is
considered to be zero, and a value of larger than one is considered to be one. Reducingfraction opt below
one will result in a reduction of the computation performed at the expense of an inferior optimal value. The
default isfraction opt = 1.0.

f 0 is a scalar variable of type defaultREAL (double precision inGALAHAD GLRT double), that gives the value of the
constant termf0 in the objective function. This value has no effect on the computed minimizerx. The default is
f 0 = 0.0.

rminvr zero is a scalar variable of type defaultREAL (double precision inGALAHAD GLRT double), that gives the
smallest value that the square of theM -norm of the gradient of the objective function may be beforeit is
considered to be zero. The default isrminvr zero = 10 u, whereu is EPSILON(1.0) (EPSILON(1.0D0) in
GALAHAD GLRT double).

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .

prefix is a scalar variable of type defaultCHARACTER and length 30, that may be used to provide a user-selected
character string to preface every line of printed output. Specifically, each line of output will be prefaced by the
stringprefix(2:LEN(TRIM(prefix))-1), thus ignoring the first and last non-null components of the supplied
string. If the user does not want to preface lines by such a string, they may use the defaultprefix = "".

2.1.2 The derived data type for holding informational parameters

The derived data typeGLRT inform type is used to hold parameters that give information about the progress and
needs of the algorithm. The components ofGLRT inform type are:

status is a scalar variable of type defaultINTEGER, that gives the current status of the algorithm. See Sections 2.3
and 2.4 for details.

alloc status is a scalar variable of type defaultINTEGER, that gives the status of the last internal array allocationor
deallocation. This will be 0 ifstatus = 0.

bad alloc is a scalar variable of type defaultCHARACTER and length 80, that gives the name of the last internal array
for which there were allocation or deallocation errors. This will be the null string ifstatus = 0.

obj is a scalar variable of type defaultREAL (double precision inGALAHAD GLRT double), that holds the value of the
objective function.

multiplier is a scalar variable of type defaultREAL (double precision inGALAHAD GLRT double), that holds the
value of the multiplierλ = σ[‖x+o‖2

M + ε]p/2−1.

leftmost is a scalar variable of type defaultREAL (double precision inGALAHAD GLRT double), that holds an estimate
of the leftmost eigenvalue of the matrix pencil(H,M).

iter is a scalar variable of type defaultINTEGER, that holds the current number of Lanczos vectors used.

iter pass2 is a scalar variable of type defaultINTEGER, that holds the current number of Lanczos vectors used in the
second pass.

negative curvature is a scalar variable of type defaultLOGICAL, that is set.TRUE. if H has been found to be
indefinite during the calculation and.FALSE. otherwise.

2.1.3 The derived data type for holding problem data

The derived data typeGLRT data type is used to hold all the data for a particular problem between calls of GLRT
procedures. This data should be preserved, untouched, fromthe initial call toGLRT initialize to the final call to
GLRT terminate.

2.2 Argument lists and calling sequences

There are three procedures for user calls (see Section 2.6 for further features):

1. The subroutineGLRT initialize is used to set default values, and initialize private data.

2. The subroutineGLRT solve is called repeatedly to solve the problem. On each exit, the user may be expected
to provide additional information and, if necessary, re-enter the subroutine.

3. The subroutineGLRT terminate is provided to allow the user to automatically deallocate array components of
the private data, allocated byGLRT solve, at the end of the solution process. It is important to do thisif the data
object is re-used for another problem sinceGLRT initialize cannot test for this situation, and any existing
associated targets will subsequently become unreachable.

We use square brackets[ ] to indicateOPTIONAL arguments.

2.2.1 The initialization subroutine

Default values are provided as follows:

CALL GLRT initialize( data, control, inform )

data is a scalarINTENT(INOUT)argument of typeGLRT data type (see Section 2.1.3). It is used to hold data about
the problem being solved.

control is a scalarINTENT(OUT)argument of typeGLRT control type (see Section 2.1.1). On exit,control con-
tains default values for the components as described in Section 2.1.1. These values should only be changed after
callingGLRT initialize.

inform is a scalarINTENT(OUT) argument of typeGLRT inform type (see Section 2.1.2). A successful call toGL-
RT initialize is indicated when the componentstatus has the value 0. For other return values ofstatus,
see Section 2.4.

2.2.2 The optimization problem solution subroutine

The optimization problem solution algorithm is called as follows:

CALL GLRT solve( n, p, sigma, X, R, VECTOR, data, control, inform[, eps, O] )

n is a scalarINTENT(IN) argument of type defaultINTEGER, that must be set to the number of unknowns,n.
Restriction: n > 0.

p is a scalarINTENT(IN) variable of type defaultREAL (double precision inGALAHAD GLRT double), that must be
set on initial entry to the desired orderp of regularisation.Restriction: p ≥ 2.

sigma is a scalarINTENT(IN) variable of type defaultREAL (double precision inGALAHAD GLRT double), that must
be set on initial entry to the value of the weightσ associated with the regularisation term.Restriction: σ ≥ 0.

X is an arrayINTENT(INOUT) argument of dimensionn and type defaultREAL (double precision inGALAHAD GL-
RT double), that holds an estimate of the solutionx of the linear system. On initial entry,X need not be set. It
must not be changed between entries. On exit,X contains the current best estimate of the solution.

R is an arrayINTENT(INOUT) argument of dimensionn and type defaultREAL (double precision inGALAHAD GL-
RT double), that is used to hold the gradientHx + c of the objective function at the current estimate of the
solution. On initial entry,R must contain the vectorc. If inform%status = 5 on exit, it must be reset toc;
otherwise it must be left unchanged. On exit,R contains the gradient of the objective function at the current best
estimate of the solution.

VECTOR is an arrayINTENT(INOUT) argument of dimensionn and type defaultREAL (double precision inGALAH-
AD GLRT double), that is used to pass information from and toGLRT solve, as explained in Section 2.3. On
initial entry, VECTOR need not be set. On exit, the actual content of the array depends on the value of the
parameterinform%status (see Section 2.3).

data is a scalarINTENT(INOUT)argument of typeGLRT data type (see Section 2.1.3). It is used to hold data about
the problem being solved. It must not have been alteredby the usersince the last call toGLRT initialize.

control is a scalarINTENT(IN) argument of typeGLRT control type. (see Section 2.1.1). Default values may be
assigned by callingGLRT initialize prior to the first call toGLRT solve.

inform is a scalarINTENT(INOUT)argument of typeGLRT inform type (see Section 2.1.2). On initial entry, the
componentstatus must be set to1. The remaining components need not be set. A successful call to GL-
RT solve is indicated when the componentstatus has the value 0. For other return values ofstatus, see
Sections 2.3 and 2.4.

eps is an optional scalarINTENT(IN) variable of type defaultREAL (double precision inGALAHAD GLRT double). If
eps is PRESENT, it must be set on initial entry to the value of the shiftε associated with the regularisation term.
Restriction: ε ≥ 0.

O is an optional arrayINTENT(IN) variable of type defaultREAL (double precision inGALAHAD GLRT double). If
O is PRESENT, it must be set on initial entry to the value of the offseto associated with the regularisation term.

2.2.3 The termination subroutine

All previously allocated arrays are deallocated as follows:

CALL GLRT terminate( data, control, inform )

data is a scalarINTENT(INOUT)argument of typeGLRT data type exactly as forGLRT solve that must not have
been alteredby the user since the last call toGLRT initialize. On exit, array components will have been
deallocated.

control is a scalarINTENT(IN)argument of typeGLRT control type exactly as forGLRT solve.

inform is a scalarINTENT(OUT)argument of typeGLRT type exactly as forGLRT solve. Only the componentstatus
will be set on exit, and a successful call toGLRT terminate is indicated when this componentstatus has the
value 0. For other return values ofstatus, see Section 2.4.

2.3 Reverse communication

A positive value ofinform%status on exit fromGLRT solve indicates that the user needs to take appropriate action
before re-entering the subroutine. Possible values are:

2. The user must perform the preconditioning operation

y := M−1z,

and recallGLRT solve. The vectorz is available in the arrayVECTOR, and the resulty must be placed inVECTOR.
No argument exceptVECTOR should be altered before recallingGLRT solve. This return can only occur when
control%unitm is .TRUE..

3. The user must perform the matrix-vector product

y := Hz

and recallGLRT solve. The vectorz is available in the arrayVECTOR, and the resulty must be placed inVECTOR.
No argument exceptVECTOR should be altered before recallingGLRT solve.

4. The user should resetR to c and recallGLRT solve. No argument exceptR should be altered before recalling
GLRT solve.

5. The user must perform the operation

y := Mz,

and recallGLRT solve. The vectorz is available in the arrayVECTOR, and the resulty must be placed inVECTOR.
No argument exceptVECTOR should be altered before recallingGLRT solve. This return can only occur when
bothcontrol%unitm is .TRUE. andO is PRESENT.

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .

2.4 Warning and error messages

A negative value ofinform%status on exit fromGLRT solve orGLRT terminate indicates that an error has occurred.
No further calls should be made until the error has been corrected. Possible values are:

-1. An allocation error occurred. A message indicating the offending array is written on unitcontrol%error, and the
returned allocation status and a string containing the nameof the offending array are held ininform%alloc -
status andinform%bad alloc respectively.

-2. A deallocation error occurred. A message indicating the offending array is written on unitcontrol%error and
the returned allocation status and a string containing the name of the offending array are held ininform%alloc -
status andinform%bad alloc respectively.

-3. (GLRT solve only) One or more of the restrictionsn > 0, sigma ≥ 0, eps ≥ 0 orp ≥ 2 has been violated.

-7. (GLRT solve only) The problem is unbounded from below. This can only happen if p ≤ 2. In this case, the
problem is unbounded along the arcX +α VECTOR asα increases.

-15. (GLRT solve only) The matrixM appears not to be positive definite.

-18. (GLRT solve only) More thancontrol%itmax iterations have been performed without obtaining convergence.

-25. (GLRT solve only) inform%status is not> 0 on entry.

2.5 Re-entry with a new value ofσ

It commonly happens that, having solved the problem for a particular value of the weightσ, a user now wishes to
solve the problem for a different value ofσ. Rather than restarting the calculation withinform%status = 1, a
useful approximation may be found resettingsigma to the new required value andR to c, and recallingGLRT solve
with inform%status = 6 and the remaining arguments unchanged. This will determinethe best solution within the
Krylov space investigated in the previous minimization (see Section 4).

2.6 Further features

In this section, we describe an alternative means of settingcontrol parameters, that is components of the variable
control of typeGLRT control type (see Section 2.1.1), by reading an appropriate data specification file using the
subroutineGLRT read specfile. This facility is useful as it allows a user to changeGLRT control parameters without
editing and recompiling programs that callGLRT.

A specification file, or specfile, is a data file containing a number of ”specification commands”. Each command
occurs on a separate line, and comprises a ”keyword”, which is a string (in a close-to-natural language) used to identify
a control parameter, and an (optional) ”value”, which defines the value to be assigned to the given control parameter.
All keywords and values are case insensitive, keywords may be preceded by one or more blanks but values must not
contain blanks, and each value must be separated from its keyword by at least one blank. Values must not contain more
than 30 characters, and each line of the specfile is limited to80 characters, including the blanks separating keyword
and value.

The portion of the specification file used byGLRT read specfile must start with a ”BEGIN GLRT” command and
end with an ”END” command. The syntax of the specfile is thus defined as follows:

( .. lines ignored by GLRT_read_specfile .. )
BEGIN GLRT

keyword value
....... .....
keyword value

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .

END
( .. lines ignored by GLRT_read_specfile .. )

where keyword and value are two strings separated by (at least) one blank. The “BEGIN GLRT” and “END” delimiter
command lines may contain additional (trailing) strings solong as such strings are separated by one or more blanks,
so that lines such as

BEGIN GLRT SPECIFICATION

and

END GLRT SPECIFICATION

are acceptable. Furthermore, between the “BEGIN GLRT” and “END” delimiters, specification commands may occur in
any order. Blank lines and lines whose first non-blank character is! or * are ignored. The content of a line after a!
or * character is also ignored (as is the! or * character itself). This provides an easy manner to ”commentout” some
specification commands, or to comment specific values of certain control parameters.

The value of a control parameters may be of three different types, namely integer, logical or real. Integer and real
values may be expressed in any relevant Fortran integer and floating-point formats (respectively). Permitted values for
logical parameters are ”ON”, ” TRUE”, ” .TRUE.”, ” T”, ” YES”, ” Y”, or ”OFF”, ” NO”, ” N”, ” FALSE”, ” .FALSE.” and ”F”.
Empty values are also allowed for logical control parameters, and are interpreted as ”TRUE”.

The specification file must be open for input whenGLRT read specfile is called, and the associated device
number passed to the routine in device (see below). Note thatthe corresponding file isREWINDed, which makes it
possible to combine the specifications for more than one program/routine. For the same reason, the file is not closed
by GLRT read specfile.

2.6.1 To read control parameters from a specification file

Control parameters may be read from a file as follows:

CALL GLRT_read_specfile( control, device )

control is a scalarINTENT(INOUT)argument of typeGLRT control type (see Section 2.1.1). Default values should
have already been set, perhaps by callingGLRT initialize. On exit, individual components ofcontrol may
have been changed according to the commands found in the specfile. Specfile commands and the component
(see Section 2.1.1) ofcontrol that each affects are given in Table 2.1.

device is a scalarINTENT(IN)argument of type defaultINTEGER, that must be set to the unit number on which the
specfile has been opened. Ifdevice is not open,control will not be altered and execution will continue, but
an error message will be printed on unitcontrol%error.

2.7 Information printed

If control%print level is positive, information about the progress of the algorithm will be printed on unitcontrol-
%out. If control%print level = 1, a single line of output will be produced for each iteration of the process. This
will include the iteration number, the value of the objective function, theM−1-norm of its gradient‖(H +λM)x+
c‖M−1, the value of the multiplierλ = σ[‖x+o‖2

M + ε]p/2−1, the number of Newton steps required to findλ, and the
exit code from this calculation (0 = successful, 1 = 1, stalled, 2 = more than 100 steps). Ifcontrol%print level ≥
2, this output will be increased to provide significant detailof each iteration. This extra output includes a complete
history of the inner iteration required to solve the “tridiagonal” subproblem, and for each Newton iteration records the
estimate ofλ, the errorθ(λ) = [‖x+o‖2

M + ε]p/2−1−λ/σ and theM -norm ofx.

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .

command component ofcontrol value type
error-printout-device %error integer
printout-device %out integer
print-level %print level integer
maximum-number-of-iterations %itmax integer
number-extra-n-vectors-used %extra vectors integer
stopping-rule %stopping rule integer
tri-diagonal-solve-frequency %freq integer
relative-accuracy-required %stop relative real
absolute-accuracy-required %stop absolute real
fraction-optimality-required %fraction opt real
constant-term-in-objective %f 0 real
zero-gradient-tolerance %rminvr zero real
two-norm-regularisation %unitm logical
space-critical %space critical logical
deallocate-error-fatal %deallocate error fatal logical

Table 2.1: Specfile commands and associated components ofcontrol.

3 GENERAL INFORMATION

Use of common: None.

Workspace: Provided automatically by the module.

Other routines called directly: GLRT solve calls the LAPACK subroutine*PTTRF, where* is S for the default real
version andD for the double precision version.

Other modules used directly: GLRT solve calls theGALAHAD packagesGALAHAD SYMBOLS, GALAHAD SPACE, GA-
LAHAD RAND, GALAHAD NORMS, GALAHAD GLTR, GALAHAD ROOTS andGALAHAD SPECFILE.

Input/output: Output is under control of the argumentscontrol%error, control%out andcontrol%print level.

Restrictions: n > 0, σ > 0.

Portability: ISO Fortran 95 + TR 15581 or Fortran 2003. The package is thread-safe.

4 METHOD

The required solutionx necessarily satisfies the optimality conditionHx+λMx +c+λo= 0, whereλ= σ[‖x+o‖2
M +

ε]p/2−1. In addition, the matrixH +λM will be positive semi-definite.

The method is iterative. Starting with the vectorM−1c, a matrix of Lanczos vectors is built one column at a time so
that thek-th column is generated during iterationk. These columns span a so-called Krylov space. The resultingn by
k matrix Qk has the property thatQT

k HQk = Tk, whereTk is tridiagonal. An approximation to the required solution
may then be expressed formally as

xk+1 = Qkyk

whereyk solves the “tridiagonal” subproblem of minimizing

1
p σ[‖y+d‖2

2+ ε]p/2+ 1
2yT Tky+ ‖c‖M−1eT

1 y, (4.1)

whered = QT
k Mo ande1 is the first unit vector.

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .

To minimize (4.1), the optimality conditions

(Tk +λI)y(λ) =−c−λd, (4.2)

whereλ = σ[‖y(λ)+d‖2
M + ε]p/2−1 are used as the basis of an iteration. Specifically, given an estimateλ for which

Tk + λI is positive definite, the tridiagonal system (4.2) may be efficiently solved to givey(λ). It is then simply a
matter of adjustingλ (for example by a Newton-like process) to solve the scalar nonlinear equation

θ(λ)≡ [‖y(λ)+d‖2
M + ε]p/2−1− λ

σ
= 0. (4.3)

In practice (4.3) is reformulated, and a more rapidly converging iteration is used.

It is possible to measure the optimality measure‖Hx + λMx + c+ λo‖M−1 without computingxk+1, and thus
without needingQk. Once this measure is sufficiently small, a second pass is required to obtain the estimatexk+1 from
yk. As this second pass is an additional expense, a record is kept of the optimal objective function values for each
value ofk, and the second pass is only performed so far as to ensure a given fraction of the final optimal objective
value. Large savings may be made in the second pass by choosing the required fraction to be significantly smaller than
one.

Special code is used in the special casep = 2, as in this case a single pass suffices.

Reference:The method is described in detail in

C. Cartis, N. I. M. Gould and Ph. L. Toint, Adaptive cubic overestimation methods for unconstrained optimization.
Technical Report RAL-TR-2007-007, Rutherford Appleton Laboratory, 2007.

5 EXAMPLE OF USE

Suppose we wish to solve a problem in 10,000 unknowns, whose data is



,

with a weightσ = 10 and shiftε = 1. Suppose further that we are content with an approximationwhich is within 99%
of the best. Then we may use the following code

PROGRAM GALAHAD_GLRT_EXAMPLE ! GALAHAD 2.4 - 15/05/2010 AT 14:30 GMT.
USE GALAHAD_GLRT_DOUBLE ! double precision version
IMPLICIT NONE
INTEGER, PARAMETER :: working = KIND( 1.0D+0 ) ! set precision
REAL ( KIND = working ), PARAMETER :: one = 1.0_working, two = 2.0_working
INTEGER, PARAMETER :: n = 10000 ! problem dimension
INTEGER :: i
REAL ( KIND = working ) :: p = 3.0_working ! order of regulatisation
REAL ( KIND = working ) :: eps = 1.0_working ! shift
REAL ( KIND = working ) :: sigma = 10.0_working ! regulatisation weight
REAL ( KIND = working ), DIMENSION( n ) :: X, R, VECTOR, H_vector, O
TYPE ( GLRT_data_type ) :: data
TYPE ( GLRT_control_type ) :: control
TYPE ( GLRT_inform_type ) :: inform
CALL GLRT_initialize( data, control, inform ) ! Initialize control parameters

control%unitm = .FALSE. ! M is not the identity matrix
control%fraction_opt = 0.99 ! Only require 99% of the best
R = one ! The linear term c is a vector of ones
O = - one ! The offset o is a vector of minus ones
inform%status = 1
DO ! Iteration to find the minimizer
CALL GLRT_solve( n, p, sigma, X, R, VECTOR, data, control, inform, &

eps = eps, O = O )
SELECT CASE( inform%status ) ! Branch as a result of inform%status
CASE( 2 ) ! Form the preconditioned vector

VECTOR = VECTOR / two ! Preconditioner is two times identity
CASE ( 3 ) ! Form the matrix-vector product

H_vector( 1 ) = - two * VECTOR( 1 ) + VECTOR( 2 )
DO i = 2, n - 1
H_vector( i ) = VECTOR( i - 1 ) - two * VECTOR( i ) + VECTOR( i + 1 )

END DO
H_vector( n ) = VECTOR( n - 1 ) - two * VECTOR( n )
VECTOR = H_vector

CASE ( 4 ) ! Restart
R = one ! set r to c

CASE( 5 ) ! Form the product of the preconditioner
VECTOR = two * VECTOR ! with a vector

CASE ( 0 ) ! Successful return
WRITE( 6, "( 1X, I0, ’ 1st pass and ’, I0, ’ 2nd pass iterations’ )" ) &
inform%iter, inform%iter_pass2

H_vector( 1 ) = - two * X( 1 ) + X( 2 )
DO i = 2, n - 1
H_vector( i ) = X( i - 1 ) - two * X( i ) + X( i + 1 )

END DO
H_vector( n ) = X( n - 1 ) - two * X( n )
WRITE( 6, "( ’ objective recurred and calculated = ’, 2ES16.8 )" ) &
inform%obj, 0.5_working * DOT_PRODUCT( X, H_vector ) + SUM( X ) + &
( sigma / p ) * ( two * DOT_PRODUCT( X + O, X + O ) + eps ) ** ( p/two )
CALL GLRT_terminate( data, control, inform ) ! delete internal workspace
EXIT

CASE DEFAULT ! Error returns
WRITE( 6, "( ’ GLRT_solve exit status = ’, I6 ) " ) inform%status
CALL GLRT_terminate( data, control, inform ) ! delete internal workspace
EXIT

END SELECT
END DO
END PROGRAM GALAHAD_GLRT_EXAMPLE

This produces the following output:

3 1st pass and 1 2nd pass iterations
objective recurred and calculated = 9.88721600E+03 9.88721600E+03

