
GALAHAD SCU
USER DOCUMENTATION GALAHAD Optimization Library version 2.5

1 SUMMARY

GALAHAD SCU is a suite of Fortran procedures for computing the thesolution to an extended system ofn+m sparse
real linear equations in n+m unknowns,

in the case where then by n matrixA is nonsingular and solutions to the systems

Ax = b and AT y = c

may be obtained from an external source, such as an existing factorization. The subroutine uses reverse communication
to obtain the solution to such smaller systems. The method makes use of the Schur complement matrix

S= D−CA−1B.

The Schur complement is stored and factorized as a dense matrix and the subroutine is thus appropriate only if there is
sufficient storage for this matrix. Special advantage is taken of symmetry and definiteness in the coefficient matrices.
Provision is made for introducing additional rows and columns to, and removing existing rows and columns from, the
extended matrix.

ATTRIBUTES — Versions: GALAHAD SCU single, GALAHAD SCU double, Uses: ROT, ROTG. Date: October
2001. Origin: N. I. M. Gould, Rutherford Appleton Laboratory, and Ph. L. Toint, University of Namur, Belgium.
Language:Fortran 95 + TR 15581 or Fortran 2003.

2 HOW TO USE THE PACKAGE

Access to the package requires aUSE statement such as

Single precision version

USE GALAHAD SCU single

Double precision version

USE GALAHAD SCU double

If it is required to use both modules at the same time, the derived typesSCU matrix type, SCU info type, and
SCU data type (Section 2.1), and the subroutinesSCU factorize, SCU solve, SCU append, SCU delete, andSCU terminate
(Section 2.2) must be renamed on one of theUSE statements.

2.1 The derived data types

Three derived data types are accessible from the package.

2.1.1 The derived data type for holding matrix information

The derived data typeSCU matrix type is used to hold data about the extended matrix and its factors. The components
of SCU matrix type are:

n is a scalar of type defaultINTEGER, that holds the value ofn, the dimension of the matrixA.

m is a scalar of type defaultINTEGER, that holds the value ofm, dimension of the matrixD.

m max is an scalar of type defaultINTEGER, that holds the dimension of the largest possible matrixD to be permitted
by the package.

class is a scalar of type defaultINTEGER, that indicates the type of matrix to be processed. Permitted values are:

1 the extended matrix is unsymmetric,

2 the extended matrix is symmetric,

3 the extended matrix is symmetric and the Schur complement matrix S is known to be positive definite.

4 the extended matrix is symmetric and the Schur complement matrix S is known to be negative definite.

BD val is a rank-one allocatable of type defaultREAL (double precision inGALAHAD SCU double), that holds the
values of the entries in the partitioned matrix

(

B
DU

)

, (2.1)

whereDU is the upper triangular part ofD, ie, (DU )i j = (D)i j if i ≤ j and zero otherwise. The entries must
be ordered by columns, with the entries in each column contiguous and those of columnj preceding those of
column j+1 ( j = 1, ....,m). The ordering within each column is unimportant.

BD row is a rank-one allocatable of type defaultINTEGER, that holds the row indices of the corresponding entries in
BD val.

BD col start is a rank-one allocatable of type defaultINTEGER, that must be set so thatBD col start(j) holds the
positions in the arraysBD val andBD row of the first entry in columnj ( j = 1, ....,m). BD col start(m+1)
must be set to the number of entries in the matrix (2) plus one.

CD val is a rank-one allocatable of type defaultREAL (double precision inGALAHAD SCU double), that need not be
set in the symmetric case. In the unsymmetric case, it holds the values of the entries in the partitioned matrix

(C DL), (2.2)

whereDL is the strict lower triangular part ofD, ie, (DL)i j = (D)i j if i > j and zero otherwise. The entries must
be ordered by rows, with the entries in each row contiguous and those of rowi preceding those of rowi+ 1
(i = 1, ....,m). The ordering within each row is unimportant.

CD col is a rank-one allocatable of type defaultINTEGER, that need not be set in the symmetric case. In the unsym-
metric case, it holds the column indices of the corresponding entries inCD val.

CD row start is a rank-one allocatable of type defaultINTEGER, that need not be set in the symmetric case. In the
unsymmetric case, it must be set so thatCD row start(i) points to the positions in the arraysCD val and
CD col of the first entry in rowi (i = 1, ....,m). CD row start(m+1) must be set to the number of entries in the
matrix (2.2) plus one.

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .

2.1.2 The derived data type for holding problem data

The derived data typeSCU data type is used to hold all the data for a particular problem between calls of SCU
procedures. This data should be preserved, untouched, fromthe initial call toSCU factorize to the final call to
SCU terminate. All components ofSCU data type are private.

2.1.3 The derived data type for holding informational parameters

The derived data typeSCU info type is used to hold parameters that give information about the factorization. The
components ofSCU info type are:

alloc status is a scalar of type defaultINTEGER, that contains the return status from the last attempted internal
workspace array allocation or deallocation. A non-zero value indicates that the allocation or deallocation was
unsuccessful, and corresponds to theSTAT= value on the user’s system. Consult local compiler documentation
for further details.

inertia is a rank-one array of length 3 and type defaultINTEGER, that holds the inertia ofSwhen the extended matrix
is symmetric. Specifically,inertia(i), i=1,2,3, give the number of positive, negative and zero eigenvalues
of S respectively.

2.2 Argument lists and calling sequences

To solve the extended system, the user must first callSCU factorize to form and factorize the Schur complement,
and then callSCU solve to compute the solution to the extended system. The solutionof additional extended systems,
with the same coefficient matrix but different right-hand sides, may be found by further calls toSCU solve.

The solution of further-extended systems of the form


may be found by firstly callingSCU append to update the existing factorization ofS (obtained fromSCU factorize)
to give that of the Schur complement ofA in the further-extended coefficient matrix and then by callingSCU solve.
Likewise, the solution of extended systems of the form

.

in which a row and column are removed from the coefficient matrix of (1.1), may be found by firstly callingSCU delete
to update the existing factorization ofS (obtained fromSCU factorize) to give that of the Schur complement ofA
in the new extended coefficient matrix and then once again by calling SCU solve.

Finally, the user may callSCU terminate to deallocate any workspace used to hold the factors ofS.

We use square brackets[ ] to indicateOPTIONALarguments.

2.2.1 The factorization stage

The Schur complement matrix may be factorized as follows:

CALL SCU factorize( matrix, data, VECTOR, status, info )

matrix is a scalarINTENT(INOUT) argument of typeSCU matrix type (see Section 2.1.1). The following compo-
nents are used bySCU factorize :

n must be set ton, the dimension of the matrixA. It is unchanged by the subroutine.Restriction: n≥ 0.

m max must be set to the dimension of the largest matrixD that will be allowed by the package. Internal
workspace of dimensionm max ∗ (m max+3)/2 (or 3m max ∗ (m max+1)/2 if class = 1) will be used by
the package. Thus care should be taken not to setm max larger than is absolutely necessary to account for
the sequence of extended matrices likely to be encountered.Restriction: m max≥ 0.

m must be set tom, dimension of the matrixD. It is unchanged by the subroutine.Restriction: 0≤ m≤ m max.

class must be set to indicate the type of matrix that will be processed (see Section 2.1.1). IfS is known to be
(positive or negative) definite throughout the sequence of extended matrices to be considered, it is more
efficient to setclass = 3 or 4. Likewise, if the matrix is known to be symmetric throughout the sequence
of extended matrices to be considered, it is more efficient tosetclass > 1. Restriction: 1≤ class≤ 4.

BD val andBD row must be set as described in Section 2.1.1 to hold the values and row indices, respectively,
of the matrix (2.1). Before use, the arrays must beALLOCATEd to be of sufficient length to hold any matrix
of the form (2.1) to be encountered in the sequence of extended matrices to be considered. Any elements
in BD val andBD row that lie below the diagonal ofD will be removed bySCU factorize.

BD col start must be set as described in Section 2.1.1 to hold the positions of the start of the columns of (2.1),
as well as to the first position past the end of the last column.Before use, this array must beALLOCATEd to
be of length at leastm max+1. The values of this array may be altered if entries that lie below the diagonal
of D are removed bySCU factorize.

CD val andCD col. Whenclass=1, these must be set as described in Section 2.1.1 to hold the values and
column indices, respectively, of the matrix (2.2). Before use, the arrays must beALLOCATEd to be of
sufficient length to hold any matrix of the form (2.2) to be encountered in the sequence of extended matrices
to be considered. Any elements inCD val andCD col that lie on or to the right of the diagonal ofD will
be removed bySCU factorize. This component need not beALLOCATEd or set ifclass > 1.

CD row start . Whenclass=1, this must be set as described in Section 2.1.1 to hold the positions of the start
of the rows in (2.2), as well as to the first position past the end of the last row. Before use, this array
must beALLOCATEd to be of length at leastm max+1. The values of this array may be altered if entries that
lie on or to the right of the diagonal ofD are removed bySCU factorize. This component need not be
ALLOCATEd or set ifclass > 1.

data is a scalarINTENT(INOUT) argument of typeSCU data type. It must not have been altered since the last call to
SCU factorize, SCU append or SCU delete.

VECTOR is anINTENT(INOUT) rank-one array argument of lengthmatrix%n and typeREAL (double precision inGA-
LAHAD SCU double), that needs not be set by the user on initial (status=1) entry. If status is greater than 1
on exit, a re-entry must be made withVECTOR set appropriately (see Section 2.3).

status is an scalarINTENT(INOUT) argument of type defaultINTEGER, that must be set by the user on initial input
to 1. On output, the value ofstatus is used to request additional information, to signal an error in the input
data or to indicate a successful call to the subroutine. A successful call is indicated by the exit valuestatus=0.
For other values, see Sections 2.3 and 2.4.

info is a scalarINTENT(INOUT) argument of typeSCU info type (see Section 2.1.3).

2.2.2 The solution stage

Solve the extended system of equations using the factorization produced by a previous call toSCU factorize,
SCU append or SCU delete, as follows:

CALL SCU solve( matrix, data, RHS, X, VECTOR, status )

matrix is a scalarINTENT(IN) argument of typeSCU matrix type (see Section 2.1.1), that must not have been
changed since the last call toSCU factorize, SCU append or SCU delete.

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .

data is a scalarINTENT(INOUT) argument of typeSCU data type. It must not have been altered since the last call to
SCU factorize, SCU append or SCU delete.

RHS is anINTENT(IN) rank-one array argument of lengthmatrix%n+matrix%m and typeREAL (double precision in
GALAHAD SCU double), that must be set on entry to the values of the right-hand-side vector

(

b1

b2

)

of the extended system of equations.RHS is not altered by the subroutine.

X is anINTENT(OUT) rank-one array argument of lengthmatrix%n+matrix%m and typeREAL (double precision in
GALAHAD SCU double). On final (status=0) exit, X contains the values of the solution

(

x1

x2

)

to the extended system of equations.

VECTOR is anINTENT(INOUT) rank-one array argument of lengthmatrix%n and typeREAL (double precision inGA-
LAHAD SCU double), that needs not be set by the user on initial (status=1) entry. If status is greater than 1
on exit, a re-entry must be made withVECTOR set appropriately (see Section 2.3).

status is an scalarINTENT(INOUT) argument of type defaultINTEGER, that must be set by the user on initial input
to 1. On output, the value ofstatus is used to request additional information, to signal an error in the input
data or to indicate a successful call to the subroutine. A successful call is indicated by the exit valuestatus=0.
For other values, see Section 2.3.

2.2.3 The updating stage

Call SCU append to extend the factorization of the Schur complement when a new row and column are appended
to the extended matrix. Subsequent systems of equations with the larger coefficient matrix may then be solved
by calls toSCU solve. Note in particular that the arraysmatrix CD val, matrix CD col, matrix CD row start,
matrix BD val, matrix BD row andmatrix BD col start must be sufficiently large to allow for the incoming row
and column. The factorization may be extended as follows:

CALL SCU append( matrix, data, VECTOR, status, info )

matrix is a scalarINTENT(INOUT) argument of typeSCU matrix type (see Section 2.1.1). The following compo-
nents are used bySCU append :

n , m max, m and class. These should not have been changed since the last call toSCU factorize,
SCU append or SCU delete. On a successful exit (status = 0) from SCU append, n will be unal-
tered andm will have been increased by 1 to account for the appended row and column. Restriction:
0≤ m≤ m max−1.

BD val andBD row. These must be set as described in Section 2.1.1 to hold the values and row indices, respec-
tively, of the further extended matrix





B c1

DU c2

d



 . (2.3)

Specifically, the firstcol start(m+1)-1 elements ofBD val andBD row should not have been changed
since the last call toSCU factorize, SCU append or SCU delete, while elementscol start(m+1), ...
should contain the values and row indices of the appended column





c1

c2

d



 ,

the ordering within this column being unimportant.

BD col start must be set as described in Section 2.1.1 to hold the positions of the start of the columns of
(2.3), as well as to the first position past the end of the last column. Specifically, the firstm+1 elements
of BD col start should not have been changed since the last call toSCU factorize, SCU append or
SCU delete, while BD col start(m+2) should be set to the number of entries in (2.3) plus one.

CD val andCD col. WhenSCU factorize was last called withmatrix%class=1, these must be set as de-
scribed in Section 2.1.1 to hold the values and column indices respectively of the further extended matrix

(

C DL

rT
1 rT

2

)

. (2.4)

Specifically, the firstrow start(m+1)-1 elements ofCD val andCD col should not have been changed
since the last call toSCU factorize, SCU append or SCU delete, while elementsrow start(m+1), ...
should contain the values and column indices of the appendedrow

(r1 rT
2 ),

the ordering within this row being unimportant. This component need not beALLOCATEd or set ifSCU fac-
torize was last called withmatrix%class > 1.

CD row start. WhenSCU factorize was last called withmatrix%class=1, this must be set as described in
Section 2.1.1 to hold the positions of the start of the rows of(6), as well as to the first position past the end
of the last row. Specifically, the firstm+1 elements ofCD row start should not have been changed since
the last call toSCU factorize, SCU append or SCU delete, while CD row start(m+2) should be set to
the number of entries in (2.4) plus one. This component need not beALLOCATEd or set ifSCU factorize
was last called withmatrix%class > 1.

data is a scalarINTENT(INOUT) argument of typeSCU data type. It must not have been altered since the last call to
SCU factorize, SCU append or SCU delete.

VECTOR is anINTENT(INOUT) rank-one array argument of lengthmatrix%n and typeREAL (double precision inGA-
LAHAD SCU double), that needs not be set by the user on initial (status=1) entry. If status is greater than 1
on exit, a re-entry must be made withVECTOR set appropriately (see Section 2.3).

status is an scalarINTENT(INOUT) argument of type defaultINTEGER, that must be set by the user on initial input
to 1. On output, the value ofstatus is used to request additional information, to signal an error in the input
data or to indicate a successful call to the subroutine. A successful call is indicated by the exit valuestatus=0.
For other values, see Section 2.3 and 2.4.

info is a scalarINTENT(INOUT) argument of typeSCU info type (see Section 2.1.3).

2.2.4 The deletion stage

Call SCU delete to extend the factorization of the Schur complement when a row and column of the existing extended
matrix are to be removed. Subsequent systems of equations with the smaller coefficient matrix may then be solved
by calls toSCU solve. The data structures for holdingB, C andD will be automatically updated to account for the
row and column deletions. Compute the factorization of the extended matrix following a row and column removal as
follows:

CALL SCU delete( matrix, data, VECTOR, status, info, col del [, row del ] )

matrix is a scalarINTENT(INOUT) argument of typeSCU matrix type (see Section 2.1.1). The following compo-
nents are used bySCU delete :

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .

n , m max, m, class, BD val, BD row, BD col start, CD val, CD col andCD row start. These should not
have been changed since the last call toSCU factorize, SCU append or SCU delete. On a successful
exit (status = 0) fromSCU delete, n will be unchanged,m will have been decreased by 1 to account for
the deleted row and column, and the contents of the arrays mayhave been changed.

data is a scalarINTENT(INOUT) argument of typeSCU data type. It must not have been altered since the last call to
SCU factorize, SCU append or SCU delete.

VECTOR is anINTENT(OUT) rank-one array argument of lengthmatrix%n and typeREAL (double precision inGALAH-
AD SCU double), that is used as workspace.

status is an scalarINTENT(OUT) argument of type defaultINTEGER. On output, the value ofstatus is used to signal
an error in the input data or to indicate a successful call to the subroutine. A successful call is indicated by the
exit valuestatus=0. For other values, see Section 2.4.

info is a scalarINTENT(INOUT) argument of typeSCU info type (see Section 2.1.3).

col del is an scalarINTENT(IN) argument of type defaultINTEGER. On input, this must be set by the user to the
index of the column of (2.1) that is to be removed.Restriction: 1≤ col del≤ m.

row del is an optional scalarINTENT(IN) argument of type defaultINTEGER. If SCU factorize was last called with
matrix%class=1, this can be set by the user to the index of the row of (2.2) thatis to be removed. If the
argument is absent, or ifmatrix%class > 1, rowcol del will be removed.Restriction: 1≤ row del≤ m.

2.2.5 The final stage

Deallocate the space required to hold the factors of the (sequence of) Schur complement(s) as follows:

CALL SCU terminate( data, status, info )

data is a scalarINTENT(INOUT) argument of typeSCU data type. It must not have been altered since the last call
to SCU factorize, SCU append or SCU delete.

status is an scalarINTENT(OUT) argument of type defaultINTEGER, that contains the exit status following a call to
SCU terminate. A successful call is indicated by the exit valuestatus=0. For other values, see Section 2.4.

info is a scalarINTENT(OUT) argument of typeSCU info type (see Section 2.1.3).

2.3 Reverse communication

If one of the subroutinesSCU factorize, SCU solve or SCU append returns with a strictly possible value ofstatus,
the subroutine in question requires the user to manipluate the vectorVECTOR, and to re-enter withstatus and all other
arguments unchanged. Possible values ofstatus and their consequences are as follows:

2 The user must obtain the solution to the system of equationsAx = b. The particular vectorb is returned in the array
VECTOR; the user must calculatex and pass this vector back inVECTOR.

3 The user must obtain the solution to the system of equationsAT y = c. The particular vectorc is returned in the
arrayVECTOR; the user must calculatey and pass this vector back inVECTOR.

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .

2.4 Warning and error messages

A negative value ofstatus on exit from any of the preceding subroutines indicates thatan error has occurred. No
further calls should be made until the error has been corrected. Possible values are:

-1 One or more of the stated restrictions on the components 1≤ matrix%class≤4,matrix%n≥ 0, 0≤ matrix%m

≤ matrix%m max, (0≤ matrix%m ≤ matrix%m max− 1 in SCU append ) 1 ≤ col del ≤ matrix%m and
1≤ row del≤ matrix%m has been violated.

-2 The subroutine has been called with an initial valuestatus ≤ 0.

-3 The factors ofS have not yet been formed indata. This indicates that eitherSCU factorize has not yet been
called, or that the last call toSCU factorize, SCU append or SCU delete ended in a failure.

-4 One or more of the arraysmatrix%BD val, matrix%BD row andmatrix%BD col start has not been allocated.

-5 When the extended matrix is unsymmetric, one or more of thearraysmatrix%CD val, matrix%CD col and
matrix%CD row start has not been allocated.

-6 One or more of the arraysmatrix%BD val, matrix%BD row andmatrix%BD col start is not large enough.
Check that the dimension ofmatrix%BD col start is no smaller thanmatrix%m+1 (matrix%m+2 for SCU app-
end ), and that those ofmatrix%BD val andmatrix%BD row are no smaller thanmatrix%BD col start(mat-
rix%m+1)-1, and re-enter. (matrix%BD col start(matrix%m+2)-1 for SCU append andmatrix%BD col st-
art(matrix%m+1)+ |matrix%col del-matrix%row del|-1 for SCU delete ).

-7 When the extended matrix is unsymmetric, one or more of thearraysmatrix%CD val, matrix%CD col and
matrix%CD row start is not large enough. Check that the dimension ofmatrix%CD row start is no smaller
thanmatrix%m+1 (matrix%m+2 for SCU append ), and that those ofmatrix%CD val andmatrix%CD col are no
smaller thanmatrix%CD row start(matrix%m+1)-1 (matrix%CD row start(matrix%m+2)-1 for SCU app-
end andmatrix%CD row start(matrix%m+1)+ |matrix%col del-matrix%row del|-1 for SCU delete ).

-8 The value recorded inmatrix m does not correspond to the dimension ofD.

-9 The Schur complement matrix is singular; this has been detected during theQR factorization ofS.

-10 The Schur complement matrix is expected to be positive definite, but this has been found not to be the case during
the Cholesky factorization ofS.

-11 The Schur complement matrix is expected to be negative definite, but this has been found not to be the case
during the Cholesky factorization of−S.

-12 An internal array allocation or deallocation failed. Seeinfo%alloc status for further details.

2.5 Multiple updates

OnceSCU append or SCU delete has been used to update the factorization of the Schur complement matrix, it is as
if the enlarged or reduced matrix were that originally factorized bySCU factorize. Consequentlysequences of row
and column additions and removals may be performed so long assufficient storage is available.

3 GENERAL INFORMATION

Use of common: None.

Workspace: Provided automatically by the module.

All use is subject to licence. Seehttp://galahad.rl.ac.uk/galahad-www/cou.html .
For any commercial application, a separate license must be signed.

8 SCU (May 8, 2013) GALAHAD



GALAHAD SCU

Other routines called directly: the basic linear algebra subprograms (BLAS)SROT andSROTG (DROT andDROTG in
GALAHAD SCU DOUBLE ) are called.

Other modules used directly: None.

Input/output: None.

Restrictions: matrix%n≥ 0, 0≤ matrix%m≤ matrix%m max and1≤ matrix%class≤ 4. Also1≤ row del≤

matrix%m and1≤ col del≤ matrix%m for SCU delete.

Portability: ISO Fortran 95 + TR 15581 or Fortran 2003. The package is thread-safe.

4 METHOD

The subroutineSCU factorize forms the Schur complementS=D−CA−1B of A in the extended matrix by repeated
reverse communication to obtain the columns ofA−1B. The Schur complement or its negative is then factorized into
its QR or, if possible, Cholesky factors.

The subroutineSCU solve solves the extended system using the following well-known scheme:

(i) Compute the solution toAu = b1;

(ii) Computex2 from Sx2 = b2−Cu;

(iii) Compute the solution toAv = Bx2; and

(iv) Computex1 = u− v.

The subroutinesSCU append andSCU delete compute the factorization of the Schur complement after a row and
column have been appended to, and removed from, the extendedmatrix, respectively. The existing factorization is
updated to obtain the new one; this is normally more efficientthan forming the factorization from scratch.

5 EXAMPLE OF USE

As a simple example, suppose we wish to solve the system of equations





















1 0 0 0 0 1 0
0 2 0 0 0 1 0
0 0 3 0 0 1 0
0 0 0 4 0 1 0
0 0 0 0 5 1 1
1 1 1 1 1 1 2
1 0 1 0 1 3 4

























































.

Notice that the leading 5 by 5 coefficient matrix is diagonal and hence easily invertible. So we might choosen = 5,
m = 2, and useSCU factorize/SCU solve to find the required solution. As the matrix is unsymmetric, we must set
matrix%class=1.

Now suppose that we have solved this system as described and are now confronted with the further system
























1 0 0 0 0 1 0 1
0 2 0 0 0 1 0 0
0 0 3 0 0 1 0 0
0 0 0 4 0 1 0 0
0 0 0 0 5 1 1 0
1 1 1 1 1 1 2 1
1 0 1 0 1 3 4 0
1 0 0 0 0 0 1 1

















































x1

x2

x3

x4

x5

x6

x7

x8

























=

























5
5
4
5
7
12
12
4

























.

Rather than applyingSCU factorize /SCU solve with n = 5, m = 3, we note that the new coefficient matrix differs
from the old one merely in having an extra row and column. Thus, we can useSCU append with n = 5, m = 2 to
update the existing factorization and then callSCU solve with n = 5, m = 3 to calculate the desired solution.

Finally, suppose that we have solved this second system and now wish to solve




















1 0 0 0 0 1 1
0 2 0 0 0 1 0
0 0 3 0 0 1 0
0 0 0 4 0 1 0
0 0 0 0 5 1 0
1 0 1 0 1 3 0
1 0 0 0 0 0 1









































x1

x2

x3

x4

x5

x6

x7





















=





















3
5
4
5
6
6
2





















.

Again, rather than applyingSCU factorize /SCU solve with n = 5, m = 2, we note that this third coefficient matrix
is the second one with itsn+1-st row andn+2-nd column removed. Thus, we can useSCU delete with n = 5, m = 3
to update the existing factorization and then callSCU solve with n = 5, m = 2 to calculate the desired solution.

To carry out these calculations, we might use the following piece of code. Notice how the extra row and column
for the second problem are simply introduced at the end of theexisting data structures.

PROGRAM GALAHAD_SCU_EXAMPLE
USE GALAHAD_SCU_DOUBLE ! double precision version
IMPLICIT NONE
INTEGER, PARAMETER :: wp = KIND( 1.0D+0 ) ! set precision
INTEGER :: i, row_del, col_del, status
INTEGER, PARAMETER :: n = 5, m = 2, mmax = m + 1
TYPE ( SCU_matrix_type ) :: mat
TYPE ( SCU_data_type ) :: data
TYPE ( SCU_info_type ) :: info
REAL ( KIND = wp ) :: X1( n + m ), RHS1( n + m )
REAL ( KIND = wp ) :: X2( n + m + 1 ), RHS2( n + m + 1 )
REAL ( KIND = wp ) :: X3( n + m ), RHS3( n + m ), VECTOR( n )
mat%m_max = mmax ; mat%class = 1
mat%n = n ; mat%m = m
ALLOCATE ( mat%BD_val( 15 ), mat%BD_row( 15 ), mat%BD_col_start( mmax+1 ),&

mat%CD_val( 13 ), mat%CD_col( 13 ), mat%CD_row_start( mmax+1 ) )
mat%BD_col_start( : 3 ) = (/1, 7, 10/)
mat%CD_row_start( : 3 ) = (/1, 6, 10/)
mat%BD_row( : 9 ) = (/1, 2, 3, 4, 5, 6, 5, 6, 7/)
mat%BD_val( : 9 ) = (/1.0_wp, 1.0_wp, 1.0_wp, 1.0_wp, &

1.0_wp, 1.0_wp, 1.0_wp, 2.0_wp, 4.0_wp/)
mat%CD_col( : 9 ) = (/1, 2, 3, 4, 5, 1, 3, 5, 6/)

mat%CD_val( : 9 ) = (/1.0_wp, 1.0_wp, 1.0_wp, 1.0_wp, &
1.0_wp, 1.0_wp, 1.0_wp, 1.0_wp, 3.0_wp/)

RHS1 = (/2.0_wp, 3.0_wp, 4.0_wp, 5.0_wp, 7.0_wp, 8.0_wp, 10.0_wp/)
RHS2 = (/5.0_wp, 5.0_wp, 4.0_wp, 5.0_wp, 7.0_wp, 12.0_wp, 12.0_wp, 4.0_wp/)
RHS3 = (/3.0_wp, 5.0_wp, 4.0_wp, 5.0_wp, 6.0_wp, 6.0_wp, 2.0_wp/)

! First system
status = 1
DO

CALL SCU_factorize( mat, data, VECTOR, status, info )
IF ( status <= 0 ) EXIT
DO i = 1, n

VECTOR( i ) = VECTOR( i ) / DBLE( FLOAT( i ) )
END DO

END DO
WRITE( 6, "( /, ’ On exit from SCU_factorize, status = ’, I3 )" ) status
IF ( status < 0 ) STOP
status = 1
DO

CALL SCU_solve( mat, data, RHS1, X1, VECTOR, status )
IF ( status <= 0 ) EXIT
DO i = 1, n ! multiply by the inverse of A

VECTOR( i ) = VECTOR( i ) / DBLE( FLOAT( i ) )
END DO

END DO
WRITE( 6, "( ’ On exit from SCU_solve, status = ’, I3 )" ) status
IF ( status < 0 ) STOP
WRITE( 6, "( /, ’ Solution (first system)’, /, ( 8ES9.2 ) )" ) X1( : )

! Second system
mat%BD_row( 10 : 12 ) = (/ 1, 6, 8 /)
mat%BD_val( 10 : 12 ) = (/ 1.0_wp, 1.0_wp, 1.0_wp /)
mat%BD_col_start( 4 ) = 13
mat%CD_col( 10 ) = 1
mat%CD_val( 10 ) = 1.0_wp
mat%CD_row_start( 4 ) = 11
status = 1
DO

CALL SCU_append( mat, data, VECTOR, status, info )
IF ( status <= 0 ) EXIT
DO i = 1, n

VECTOR( i ) = VECTOR( i ) / DBLE( FLOAT( i ) )
END DO

END DO
WRITE( 6, "( /, ’ On exit from SCU_append, status = ’, I3 )" ) status
IF ( status < 0 ) STOP
status = 1
DO

CALL SCU_solve( mat, data, RHS2, X2, VECTOR, status )
IF ( status <= 0 ) EXIT
DO i = 1, n

VECTOR( i ) = VECTOR( i ) / DBLE( FLOAT( i ) )
END DO

END DO
WRITE( 6, "( ’ On exit from SCU_solve, status = ’, I3 )" ) status
IF ( status < 0 ) STOP
WRITE( 6, "( /, ’ Solution (second system)’, /, ( 8ES9.2 ) )" ) X2

! Third system
row_del = 1
col_del = 2
status = 1
CALL SCU_delete( mat, data, VECTOR, status, info, col_del, &

row_del = row_del )
WRITE( 6, "( /, ’ On exit from SCU_delete, status = ’, I3 )" ) status
IF ( status < 0 ) STOP
status = 1
DO
CALL SCU_solve( mat, data, RHS3, X3, VECTOR, status )
IF ( status <= 0 ) EXIT

DO i = 1, n
VECTOR( i ) = VECTOR( i ) / DBLE( FLOAT( i ) )

END DO
END DO
WRITE( 6, "( ’ On exit from SCU_solve, status = ’, I3 )" ) status
IF ( status < 0 ) STOP
WRITE( 6, "( /, ’ Solution (third system)’, /, ( 8ES9.2 ) )" ) X3
CALL SCU_terminate( data, status, info )
WRITE( 6, "( /, ’ On exit from SCU_terminate, status = ’, I3 )" ) status
END PROGRAM GALAHAD_SCU_EXAMPLE

This produces the following output:

On exit from SCU_factorize, status = 0
On exit from SCU_solve, status = 0

Solution (first system)
1.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00

On exit from SCU_append, status = 0
On exit from SCU_solve, status = 0

Solution (second system)
3.00E+00 2.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00

On exit from SCU_delete, status = 0
On exit from SCU_solve, status = 0

Solution (third system)
1.00E+00 2.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00

On exit from SCU_terminate, status = 0

